
Open Problem for SUAVe User Group Meeting, November 3rd, 2015 
 

Consider the following SAS datastep:  

 
data fake; 

 input subject $ rating meas_year subj_age; 

 cards; 

 Aaron  48  2006 22 

 Aaron  101  2010 26 

 Aaron  112  2014 . 

 Peyton 99  2003 27 

 Peyton 98  2007 . 

 Peyton 106  2012 . 

 Tony  95  2006 26  

 Tony  95  2010 30 

 Tony  113  2014 34 

 Tom  87  2001 . 

 Tom  88  2006 29 

 Tom  111  2010 .  

 ;  

run; 

 

Each subject has been repeatedly measured every 5 or so years. In the dataset, the measurement year (meas_yr) is 

always present but the subject’s age (subj_age) is often missing. Luckily both values exist together at least once for each 

subject.  

 

The challenge: write a SAS program to convert the missing ages to correct values based on knowing the measurement 

year. For simplicity assume that yearly measurements are always taken on the same calendar date. 

Bonus: what if there were two inconsistent ages for the same subject.  For instance, in the first observation, what if 22 

was replaced by 21 - how would you handle this? 

 

If you'd like to share your solution at the SUAVe meeting, please send it to one of the SUAVe executive anytime between 

now and October 26th, 2015. The (first) 3 most imaginative solutions will be given preference.  

 

Good luck!  

 

Cheryl Sivertson 

Mike Atkinson 

Joel Choy  

Steve Gibbs 

Peter Ott 

  



Solutions 

 

Joel Choy 

 
libname curr 'H:\SAS User group\Oct162015\'; 

 

data curr.fake; 

 input subject $ rating meas_year subj_age; 

 cards; 

 Aaron  48  2006 22 

 Aaron  101  2010 26 

 Aaron  112  2014 . 

 Peyton  99  2003 27 

 Peyton  98  2007 . 

 Peyton  106  2012 . 

 Tony  95  2006 26  

 Tony  95  2010 30 

 Tony  113  2014 34 

 Tony       113     2015    34  

 Tom  87  2001 . 

 Tom  88  2006 29 

 Tom  111  2010 .  

 ;  

run; 

 

data curr.fake01; 

 set curr.fake; 

 if subj_age not = . 

 then birth_dt = meas_year - subj_age;  

run;    

 

proc sql; 

 create table curr.birthdt as  

 select distinct  subject, birth_dt from curr.fake01 

 where birth_dt not =.;     

quit;  

 

* remove duplicate subject with multiple birthdates; 

* keep the birthdate that is oldest date; 

 

/* proc sort nodupkey is simplest but would not allow you to choose which to remove */ 

 

proc sql; 

 create table curr.birthdts as 

 select *, freq(subject) as num_subject 

 from curr.birthdt group by subject having  num_subject = 1 

 union 

 select subject, min(birth_dt), freq(subject) as num_subject 

 from curr.birthdt group by subject having num_subject > 1;   

quit;  

 

proc sql; 

 create table curr.fake02 as 

 select t1.subject, t1.rating, t1.meas_year, (t1.meas_year - t2.birth_dt) as subj_age  

 from curr.fake01 t1, curr.birthdts t2 

 where t1.subject = t2.subject; 

quit;  

 

  

 

 



Jim Goudie 

 
data fake; 

 input subject $ rating meas_year subj_age; 

 if subj_age ^=. then diff=meas_year-subj_age; 

 cards; 

 Aaron  48  2006 21 

 Aaron  101  2010 26 

 Aaron  112  2014 . 

 Peyton  99  2003 27 

 Peyton  98  2007 . 

 Peyton  106  2012 . 

 Tony  95  2006 26  

 Tony  95  2010 30 

 Tony  113  2014 34 

 Tom  87  2001 . 

 Tom  88  2006 29 

 Tom  111  2010 .  

 ;  

run; 

 

proc sort data=fake; 

 by subject meas_year; 

 

Proc means data=fake noprint; 

 by subject; 

 var diff; 

 output out=diff_means mean=; 

 

data fake2 (drop=diff  ); 

 merge fake (drop=  diff)  

    diff_means(keep=subject diff); 

 by subject ; 

 *if conflicts exist, take the mean diff rounded to the nearest number of years (bonus 

problem); 

 if subj_age=. then do; 

 subj_age=meas_year-round(diff); 

 flag="<--------predicted"; 

end; 

 

proc print; 

run; 

 

quit; 

  



Gord Nigh 

 
data fake; 

input subject $ rating meas_year subj_age; 

if subject = 'Aaron' then name =       'Aaron Rodgers              '; 

else if subject = 'Peyton' then name = 'Peyton Manning'; 

else if subject = 'Tony' then name =   'Tony "Choke" Romo'; 

else if subject = 'Tom' then name =    'Tom "The Deflator" Brady'; 

else if subject = 'Brett' then name =  'Brett Favre'; 

else if subject = 'Cam' then name =    'Cam "Not old enough" Newton'; 

cards; 

 Aaron  48  2006 22 

 Aaron  101  2010 26 

 Aaron  112  2014 . 

 Peyton  99  2003 27 

 Peyton  98  2007 . 

 Peyton  106  2012 . 

 Tony  95  2006 26  

 Tony  95  2010 30 

 Tony  113  2014 34 

 Tom  87  2001 . 

 Tom  88  2006 29 

 Tom  111  2010 .  

 Cam  95  2006 .  

 Cam  95  2010 . 

 Cam  113  2014 . 

 Brett  87  2001 13 

 Brett  88  2006 17 

 Brett  111  2010 21  

 ;  

run; 

 

proc sort data=fake; 

 by subject subj_age; 

 

data birth_year; 

 set fake; 

 if subj_age = . then delete; else birth_year = meas_year - subj_age; 

 

data birth_year; 

 set birth_year; 

 retain year; 

 by subject; 

 if first.subject then year = birth_year; 

 else if year ne birth_year then year = .; 

 if last.subject then output; 

 keep subject year; 

 

data fake; 

 merge fake birth_year; 

 by subject; 

 comment = '           '; 

 if year = . then do; subj_age = .; comment = 'Unknown age'; end; 

 else subj_age = meas_year - year; 

 

proc sort data=fake; 

 by subject meas_year; 

 

proc print;  

 var name rating meas_year subj_age comment; 

run;  

quit; 



Steve Gibbs 

 
**********************************************; 

* Steve Gibbs Open Problem Solution Nov 2015 *; 

**********************************************; 

data fake; 

 input subject $ rating meas_year subj_age; 

 cards; 

 Aaron  48  2006 22 

 Aaron  101  2010 26 

 Aaron  112  2014 . 

 Peyton 99  2003 27 

 Peyton 98  2007 . 

 Peyton 106  2012 . 

 Tony  95  2006 26  

 Tony  95  2010 30 

 Tony  113  2014 34 

 Tom  87  2001 . 

 Tom  88  2006 29 

 Tom  111  2010 .  

 ;  

run; 

 

* Part 1: Find age for missing records *; 

%macro mop1(dsnin1=,dsnout1=,m=); 

proc sort data=&dsnin1 out=&m.1; 

 by subject descending subj_age; 

run; 

 

data &dsnout1; 

 attrib yob length=8. label='Year of Birth' 

        subj_age_full length=8 format=3.0 label='Subject Age (no missing values)'; 

 retain yob; 

 set &m.1; 

 by subject descending subj_age; 

 if first.subject then yob=. ; 

 if nmiss(subj_age,meas_year)=0 then do; 

  yob=(meas_year-subj_age); 

  subj_age_full=subj_age; 

 end; 

 else if subj_age = . then subj_age_full=(meas_year-yob); 

run; 

 

proc sort data=&dsnout1;  

 by subject meas_year; 

run; 

%mend mop1; 

 

%mop1(dsnin1=fake,dsnout1=op_fake,m=op1); 

 

title 'Open Problem Part 1'; 

proc print data=op_fake noobs width=min; 

 var subject rating meas_year yob subj_age subj_age_full; 

run; 

 

* Part 2:  Create a var called DATAPROB to identify Subs with >1 Year of Birth (YOB) 

Value *; 

* Records with DATAPROB=1 may be excluded for subsequent analysis depending on importance 

of age. *; 

* If a value for age is needed then imputation could be used eg Last Observation Carried 

Forward. *; 

data fake2; 



 set fake; 

 if subj_age=22 then subj_age=21; 

run; 

 

%mop1(dsnin1=fake2,dsnout1=op_fake2,m=op2); 

 

proc sort data=op_fake2 out=pt2a; 

 by subject yob; 

run; 

 

data pt2b (keep=subject datchk); 

 length datchk 8. ; 

 retain datchk ; 

 set pt2a; 

 by subject yob; 

 if first.subject then datchk=.; 

 if first.yob then datchk+1; 

 if last.subject; 

run; 

 

data pt2c; 

 merge pt2a pt2b (where=(datchk gt 1)); 

 by subject; 

run; 

 

data pt2; 

 length dataprob 8. ; 

 set pt2c; 

 if datchk gt 1 then do; 

  dataprob=1; 

  if subj_age = . then do; 

   yob = . ; 

   subj_age_full = . ; 

  end; 

 end; 

 else dataprob=0; 

run; 

 

proc sort data=pt2; 

 by subject meas_year; 

run; 

 

title 'Open Problem Part 2'; 

proc print data=pt2 noobs width=min; 

 var subject rating meas_year yob subj_age subj_age_full dataprob; 

run; 

 

* If a value for age is needed then imputation could be used eg Last Observation Carried 

Forward. *; 

data pt3; 

 length yob2 8. ; 

 retain yob2 ; 

 set pt2; 

 by subject meas_year; 

 if first.subjid then yob2=.; 

 if yob ne . then yob2=yob; 

 if subj_age_full = . then subj_age_full=(meas_year-yob2); 

run; 

 

title 'Open Problem Part 2 LOCF'; 

proc print data=pt3 noobs width=min; 

 var subject rating meas_year yob yob2 subj_age subj_age_full dataprob; 

run; 



Dale Starr 
 

/* 

 * Original Data for Open Problem; 

 DATA fake; 

  INPUT subject $ rating meas_year subj_age; 

  CARDS; 

 Aaron 48 2006 21 

 Aaron 101 2010 26 

 Aaron 112 2014 . 

 Peyton 99 2003 27 

 Peyton 98 2007 . 

 Peyton 106 2012 . 

 Tony 95 2006 26  

 Tony 95 2010 30 

 Tony 113 2014 34 

 Tom 87 2001 . 

 Tom 88 2006 29 

 Tom 111 2010 .  

 ;  

 RUN; 

 */ 

 

 * Modified data for Open Problem; 

 DATA fake; 

  INPUT subject $ rating meas_year subj_age; 

  CARDS; 

 Aaron 18 1994 10 

 Aaron 28 1998 . 

 Aaron 38 2002 20 

 Aaron 48 2006 21 

 Aaron 101 2010 26 

 Aaron 112 2014 . 

 Aaron 132 2018 37 

 Peyton 99 2003 27 

 Peyton 98 2007 . 

 Peyton 106 2012 . 

 Tony 95 2006 26  

 Tony 95 2010 30 

 Tony 113 2014 34 

 Tom 87 2001 . 

 Tom 88 2006 29 

 Tom 111 2010 .  

 ;  

 RUN; 

 

 /*  

  Calculate birth year if subject age is given 

  and calcualte the frequency of each birth year. 

 

  Additional variables created 

    meas_yr_minus_age_brth_yr - the calculated birth year, meas_year - subj_age 

    individual_year_count     - number of times a value of meas_yr_minus_age_brth_yr 

appears in the data for a subject 

    total_year_count          - number of entries for a subject for which a 

meas_yr_minus_age_brth_yr can be calculated 

                                i.e. an age is given for the subject 

    max_birth_year            - maxminum value of the calculated birth year for a subject 

    min_birth_year            - minimum value of the calculted birth year for a subject 

 

  The sorting of the data in the ORDER BY statement determines which birth year is used 

  to calculate the missing age in the next program step. 

 */ 



 PROC SQL; 

   CREATE TABLE fake_vers02 AS 

    SELECT b.subject, 

           b.rating, 

           b.meas_year, 

           b.subj_age AS given_subj_age, 

           b.meas_yr_minus_age_brth_yr, 

           CASE WHEN b.meas_yr_minus_age_brth_yr NE . THEN 

FREQ(b.meas_yr_minus_age_brth_yr) 

          ELSE . END AS individual_year_count, 

           b.total_year_count,           

           b.max_birth_year, 

           b.min_birth_year 

     FROM ( 

           SELECT a.subject, 

                  a.rating, 

                  a.meas_year, 

                  a.subj_age, 

                  CASE WHEN a.subj_age NE . THEN a.meas_year - a.subj_age 

                 ELSE . END AS meas_yr_minus_age_brth_yr, 

                  MAX(CASE WHEN a.subj_age NE . THEN a.meas_year - a.subj_age 

                      ELSE . END) AS max_birth_year, 

                  MIN(CASE WHEN a.subj_age NE . THEN a.meas_year - a.subj_age  

                      ELSE . END) AS min_birth_year, 

                  CASE WHEN subj_age NE . THEN FREQ(a.subj_age) 

                       ELSE FREQ(a.subj_age) END AS total_year_count 

             FROM fake a 

/* Note the GROUP BY statement is NOT standard SQL.  */ 

/* SAS allows the variables listed in the GROUP BY statement */ 

/* to be a subset of the variables listed in the SELECT statement. */ 

/* The MIN and MAX functions in are only applied against the variable */ 

/* "subject", not unlike how a "PARTITION BY" statement would work in standard SQL*/ 

         GROUP BY a.subject 

     ) b 

/* The FREQ function in the CASE statement is only against the two */ 

/* variables "subject", and "meas_yr_minus_age_brth_yr", not unlike */ 

/* how a "PARTITION BY" statement would work in standard SQL */ 

 GROUP BY b.subject, 

          b.meas_yr_minus_age_brth_yr 

 ORDER BY b.subject, 

          missing(b.meas_yr_minus_age_brth_yr), 

          individual_year_count, 

          b.meas_yr_minus_age_brth_yr, 

          b.meas_year; 

 QUIT; 

    

 /*  

  Calculate age if missing using the most frequent 

  birth year if applicable, or using the only birth year 

  available, or if there is no indication of a most probable 

  birth year then the most recently available birth year. 

 

  To distinguish between age values that were in the original 

  data set and calculated age values, meas_year - meas_yr_minus_age_brth_yr, 

  the calculated age is stored in variable "calc_subj_age" and the value 

  of meas_yr_minus_age_brth_yr used in the calculation is stored in 

"brth_yr_used_in_age_calc". 

 */ 

 

 DATA fake_vers03 (drop = temp_birth_yr); 

    SET fake_vers02; 

 BY subject; 

 



    RETAIN temp_birth_yr temp_meas_year; 

    IF first.subject THEN DO; 

      temp_birth_yr = meas_yr_minus_age_brth_yr; 

   brth_yr_used_in_age_calc = meas_yr_minus_age_brth_yr; 

    END; 

 

    ELSE DO; 

      IF MISSING(given_subj_age) THEN DO; 

        calc_subj_age = meas_year - temp_birth_yr; 

        brth_yr_used_in_age_calc = temp_birth_yr; 

      END; 

      ELSE DO; 

        brth_yr_used_in_age_calc = meas_yr_minus_age_brth_yr; 

        calc_subj_age = .; 

        temp_birth_yr = meas_yr_minus_age_brth_yr; 

      END; 

    END; 

 RUN; 

 

 /*  

  Reduce fields and resort data for final output. 

 

  The variable "max_age_error_rng_used_vs_MinMax" is the maximum  

  differential between the "brth_yr_used_in_age_calc" and the  

  max and min values of the calculted birth year for a subject. 

 

  The final data set lets the user know what varibles were in the 

  original data, 

    Name 

    subject 

    rating 

    meas_year 

    given_subj_age 

  what variables were calculated, 

    calc_subj_age 

    meas_yr_minus_age_brth_yr 

  what values were used in intermedidate calculations. 

    brth_yr_used_in_age_calc 

  and the the possible error range is on calculated values 

    max_age_error_rng_used_vs_MinMax 

 */ 

 PROC SQL; 

   CREATE TABLE fake_vers04 AS 

     SELECT a.subject, 

            a.rating, 

            a.meas_year, 

            a.given_subj_age, 

            a.calc_subj_age,             

            a.meas_yr_minus_age_brth_yr, 

            a.brth_yr_used_in_age_calc, 

            CASE WHEN a.total_year_count > 1 THEN "+/-" || 

TRIM(PUT(MAX(ABS(a.max_birth_year - a.brth_yr_used_in_age_calc), ABS(a.min_birth_year - 

a.brth_yr_used_in_age_calc)), 2.)) 

          ELSE "indeterminate" END AS max_age_error_rng_used_vs_MinMax 

       FROM fake_vers03 a 

   ORDER BY subject, 

            meas_year; 

 QUIT; 

 


